

TODO SOBRE EL

AMSTRAD

AÑO 1 • N° 4

420 pts.
10
PROGRAMAS

- CABECERA
- ARCHIVO
- CONDOR
- DEMOLER
- ESPACIO
- FACTURAS
- GRAFICAS
- LUNA - 7
- PING - PONG
- SUB - 85
- TANK 2000





LO ÚNICO...

UN CHORRO

EFICAZ

PARA UNA BUENA CARGA
DEL PROGRAMA EN TU ORDE-
NADOR, COMPRUEBE
QUE LA CABEZA DE SU
PLATINA ESTE
LIMPIA Y EL
AZIMUTH PERFECTA-
MENTE AJUSTADO.
LIMPIELA CON EL
PRODUCTO
SKUA.



LIMPIADOR DE RADIO CASSETTES



FABRICADO POR

ASKLE
CHEMICAL S.A.
ALICANTE (SPAIN)

BOLETIN DE PEDIDO

Nombre y Apellidos


Dirección

Ciudad Telf. D.P.

Deseo recibir..... Limpiadores de cassettes al precio de 420 Ptas. Unidad.

Forma de pago: ☐ Talón ☐ Contrareembolso ☐ Giro postal

Remite a GTS S.A. Boilón nº 20 1.ª Izda. D.P. 28005



EDITA:

EDITORIAL GTS. Bailén, 20
1.º Izq. Tlf. 266 66 01-02
28005 MADRID

SECRETARIA REDACCION:

N. Vera Clavijo.

COLABORADORES:

Eugenio Garrido.
J. F. Martínez.
J. Bernal.
R. Carralón.
J. Ramos.
Juan Jesús Ortega.

**DIRECCION ARTISTICA
Y TECNICA:**

Jesús Negrete.

PUBLICIDAD:

Dpto. propio:
Bailén, 20-1.º Izq.
28005 MADRID

FOTOCOMPOSICION:

Ⓜ errata, S. A.
C/. Alejandro González, 7
28028 MADRID

IMPRIME:

Gráficas Futura.


PRODUCCION CASSETTES:

Iberofón, S. A.

DISTRIBUYE:

R.B.A. Promotora de Ediciones,
S. A. Trav. de Gracia, 56. Atico
1.º. Tfno.: 200 82 56.

Depósito Legal: AV-266-1985



EDITORIAL

Queridos lectores:

Los programas que hemos diseñado para este número los hemos enfocado en dos vertientes. Una educacional, para que cada vez conozcas mejor tu aparato, y la otra para que al mismo tiempo te resulte divertido.


Por lo que respecta a la revista, como verás, algunos de los programas los hemos listado. Esto es porque suponemos que el lector tiene ya cierto conocimiento de las órdenes principales de su aparato. Intentamos de esta forma que el usuario se familiarice, mejor sus conocimientos y al mismo tiempo se divierta.

Ninguno de los programas es excesivamente largo. Ya iremos ampliándolos progresivamente.

También hemos intentado que la mayoría de ellos tengan un elemento de juego, para animaros a competir con la máquina.

Por otra parte, cuentas con una página de bibliografía, con las últimas novedades del mercado editorial sobre tu ordenador.


Hasta el próximo mes, que te diviertas.




EXPLICACION DE LOS PROGRAMAS

Pág. **4**

LIBROS



Pág. **18**



EXPLICACION DE LOS PROGRAMAS

GRAFICAS

Este es un ejemplo de representación gráfica de funciones para valores de X comprendidos entre $X = -5$ y $X = +5$.

Para ilustrarlo hemos introducido las ecuaciones $y = \cos(x)$ y $h = \tan(x)$. Si bien tu puedes introducir cualquier otra. Pero recuerda que has de hacerlo siguiendo la terminología Basic-Microsoft, y que no debes modificar ningún dato a la izquierda de ambas igualdades.


```
10 INK 0,1:INK 1,26:INK 2,6:INK 3,2,8:PAPER 0
20 MODE 0:BORDER 1:LOCATE 1,12:PEN 3:PRINT"          GRAFICAS      ":FOR t=10 TO 500 ST
EP 10:SOUND 5,t,10:NEXT
30 MODE 1:INK 0,26:INK 1,0
40 PRINT:PRINT" Este es un ejemplo de representacion"
50 PRINT:PRINT" grafica de funciones para valores de"
60 PRINT:PRINT"      x comprendidos entre x=-5 y x=+5."
70 PRINT:PRINT:PRINT"      Para ilustrarlo hemos introducido"
80 PRINT:PRINT"      las ecuaciones y=cos(x) y h=tan(x),"
90 PRINT:PRINT"      " si bien, tu puedes representar cual-"
100 PRINT:PRINT"      " quier otra. pero recuerda has de ha-"
110 PRINT:PRINT"      " cerlo segun la terminologia BASIC-"
120 PRINT:PRINT" MICROSOFT y no debes modificar nin-"
130 PRINT:PRINT"      gun dato a la izquierda de ambas i- "
140 PRINT:PRINT"      gualdades.
150 LOCATE 1,25:PRINT"      PULSA UNA TECLA PARA CONTINUAR "
160 IF INKEY$="" THEN 160 ELSE CLS
170 KEY 128,"run 170"+CHR$(13)
180 MODE 0:BORDER 4
190 INK 0,26:INK 1,0
200 DRAW 320,0
210 DRAW 320,400:DRAW 320,0
220 MOVE 0,200
230 DRAW 640,200
240 MOVE 1,197:DRAW 1,204
250 MOVE 64,197:DRAW 64,204
260 MOVE 128,197:DRAW 128,204
270 MOVE 192,197:DRAW 192,204
280 MOVE 256,197:DRAW 256,204
290 MOVE 384,197:DRAW 384,204
300 MOVE 448,197:DRAW 448,204
310 MOVE 512,197:DRAW 512,204
320 MOVE 576,197:DRAW 576,204
330 MOVE 639,197:DRAW 639,204
340 PLOT 325,8:DRAW 317,8:DRAW 317,8
```



```

350 PLOT 325,72:DRAW 317,72
360 PLOT 325,136:DRAW 317,136
370 PLOT 325,264:DRAW 317,264
380 PLOT 325,328:DRAW 317,328
390 PLOT 325,392:DRAW 317,392
400 FOR x=-5 TO 5 STEP 0.02
410 IF x=0 GOTO 530
420 y=COS(x):h=TAN(x)
430 IF Y<-3.5 THEN Y=-3.5
440 IF Y>3.5 THEN Y=3.5
450 IF H<-3.5 THEN H=-3.5
460 IF H>3.5 THEN H=3.5
470 j=320+(X*64)
480 i=200+(Y*64)
490 m=200+(H*64)
500 PLOT j,i,3
510 PLOT j,m,1
520 FRAME
530 NEXT
540 c=1:FOR i=1 TO 2500:NEXT
550 MODE 1
560 IF c=0 THEN 630
570 LOCATE 1,12:PRINT"QUIERES REPRESENTAR OTRAS FUNCIONES S/N"
580 a$=INKEY$
590 IF a$="S" OR a$="s" THEN 630
600 IF a$="N" OR a$="n" THEN 620
610 GOTO 580
620 CLS:END
630 MODE 1
640 PRINT:PRINT:PRINT"      Una vez intoducidas las nuevas
                                funciones a representar pul
sa
te- "                                ENTER y seguidamente en el
650 PRINT:PRINT"                  clado numerico f0."
660 LOCATE 1,18:EDIT 420

```



LUNA-7


Con tu nave espacial debes intentar alunizar usando con precaución tus retropropulsores de modo que no te quedes sin combustible demasiado lejos de la superficie lunar ya que caerías atraído por la fuerza de la gravedad de la luna. Pero tampoco por economizar vayas a hacerle un agujero a la luna.

```

10 SYMBOL 250,&X1000,&X1000,&X11100,&X110110,&X101010,&X101010,&X1110111,&X111111
0
20 SYMBOL 251,&X10100,&X10100,&X110110,&X100010,&X1110111,&X1111111,&X1100011,&X
1000001
30 SYMBOL 252,&X10000,&X0,&X101000,&X0,&X1010100,&X0,&X10101010
40 INK 0,13:INK 1,26:INK 13,6:INK 3,3,8:PAPER 0
50 MODE 0:BORDER 1:LOCATE 1,12:PEN 3:PRINT"      LUNA-7      ":FOR t=10 TO 500 STE
P 10:SOUND 5,t,10:NEXT
60 MODE 1:PEN 1:PRINT:PRINT:PRINT"      TE ENCUENTRAS EN EL ESPACIO A "
70 PRINT:PRINT"      BORDO DE UNA NAVE QUE VIAJA"
80 PRINT:PRINT"      HACIA LA LUNA.DEBES ALUNIZAR"
90 PRINT:PRINT"      LO MAS DESPACIO QUE PUEDES"

```


CINTA "TSA YTB
DISCO 1CB "LUNA7" SK




```

100 PRINT:PRINT " UTILIZANDO TUS RETROPROPULSORES."
110 PRINT:PRINT:PRINT:PRINT " !! BUENA SUERTE !!"
120 LOCATE 1,20:PRINT" PARA COMENZAR EL JUEGO PULSA UNA TECLA"
130 IF INKEY$="" THEN 130 ELSE CLS
140 MODE 1:BORDER 13
150 INK 0,13:INK 1,0
160 DIM f(30), a(30), c(30), v(30), r(30)
170 CLS
180 DEF FNa(x)=INT(x*100)/100
190 f1=0:a=70:v=1:f2=32500:f3=16500
200 g=0.001:z=1.8
210 GOSUB 1030
220 LOCATE 1,1:PRINT "TIEMPO ALTURA VELOCIDAD COMBUST CONSUMO"
230 LOCATE 1,2:PRINT " SEG. METROS KM/H LITROS LIT/SEG
240 IF A=70 THEN 320
250 U=U+1
260 IF u>9 THEN B=B+1
270 FOR j=b TO u-1
280 PRINT:PRINT TAB(1);f(j);TAB(7);a(j);
290 PRINT TAB(16);v(j);TAB(25);r(j);
300 PRINT TAB(33);"C=";c(j)
310 NEXT
320 PRINT:PRINT TAB(1);INT(f1+0.5);TAB(7);INT(a*1000);TAB(16);FNa(v*3600);TAB(25);FNa(f2-f3);
330 a(u)=INT(a*1000);f(u)=INT(f1+0.5):V(u)=FNa(v*3600)
340 r(u)=FNa(f2-f3)
350 PRINT TAB(33);"C=";
360 LINE INPUT c$:c=VAL(c$)
370 IF c<0 THEN c=0
380 IF c>500 THEN PRINT "*** MAXIMA POTENCIA=500 LITROS/SEG. ***":PRINT:GOTO 350
390 C(U)=C
400 T=10
410 IF C=0 THEN 420
420 IF F2-F3-0.001<0 THEN 530
430 IF T-0.001<0 THEN 210
440 S=T
450 IF F3+S*C-F2<=0 THEN 470
460 S=(F2-F3)/C
470 GOSUB 980
480 IF F4<=0 THEN 810
490 IF V<=0 THEN 510
500 IF F5<0 THEN 860
510 GOSUB 930
520 GOTO 420
530 CLS:LOCATE 1,10:PRINT"***** FIN DEL COMBUSTIBLE A LOS";FNA(F1);"SEG. DE INICIADA LA MANIOBRA Y A"; INT(A*1000);"METROS DE ALTITUD SOBRE LA LUNA *****"
540 FOR I=1 TO 5000:NEXT I:A=0
550 s=(-v+SQR(v*v+2*a*g))/g
560 v=v+g*s
570 f1=f1+s
580 GOSUB 1030
590 CLS:LOCATE 1,10:PRINT"**** SOBRE LA LUNA A LOS";FNA(F1);"SEG ****"
600 W=3600*V
610 PRINT:PRINT"** VELOCIDAD DE IMPACTO=";INT(W);"KM/H **"
620 PRINT:PRINT"** RESERVA DE COMBUSTIBLE=";INT(FNA(F2-F3));"LTS **"
630 IF W>5 THEN 660
640 PRINT:PRINT"**** TU ALUNIZAJE HA SIDO PERFECTO ****"
650 GOTO 760
660 IF W>20 THEN 690
670 PRINT:PRINT "*** HAS REALIZADO UN BUEN ALUNIZAJE ***"
680 GOTO 760
690 IF W>55 THEN 720
700 PRINT:PRINT "*** HAS REALIZADO UN MAL ALUNIZAJE ***"
710 GOTO 760
720 IF W>85 THEN 750
730 PRINT:PRINT"** HAS DESTROZADO LA NAVE !!INUTIL!!** "
740 GOTO 760
750 PRINT:PRINT"***** TE HAS ESTRELLADO CREANDO UN NUEVO CRATER LUNAR DE ";FNA(W*0.15);"METROS *****"
760 PRINT:PRINT" ***** LO INTENTAS DE NUEVO ;S/N¿ *****"
770 A$=INKEY$
780 IF A$="S" OR A$="s" THEN CLEAR: GOTO 140
790 IF A$="N" OR A$="n" THEN CLS :END
800 GOTO 770
810 IF s-0.005<0 THEN 580
820 s=2*a/(v+SQR(v*v+2*a*(g-z*c/f2)))
830 GOSUB 980
840 GOSUB 930
850 GOTO 810

```

```
860 w=(1-f2*g/(c*z))/2
870 s=f2*v/(z*c*(w+SQR(w*w+v/z)))+0.05
880 GOSUB 980
890 IF f4<=0 THEN 810
900 GOSUB 930
910 IF f5>=0 THEN 420
920 IF v<=0 THEN 420 ELSE GOTO 860
930 f1=f1+s
940 t=t-s
950 f2=f2-s*c
960 a=f4:v=f5
970 RETURN
980 q=s*c/f2
990 IF q>0 AND q<3.40001E-08 THEN q=3.40001E-08
1000 f5=v+g*s+z*(-q-q^2/2-q^3/3-q^4/4-q^5/5)
1010 f4=a-g*s^2/2-v*s+z*s*(q/2+q^2/6+q^3/12+q^4/20+q^5/30)
1020 RETURN
1030 IF a<>70 THEN 1040
1040 MODE 1:BORDER 0:INK 0,0:INK 1,26:INK 2,11:INK 8,15:DEG:TAG
1050 FOR I=1 TO 640 STEP 8
1060 J=RND*400
1070 PLOT I,J,1
1080 NEXT
1090 IF A<=0 THEN A=0.1
1100 X=(320/(8*A))+4
1110 y=3.25*(INT(LOG(a+0.5)*70))-575
1120 IF a>10 THEN E=(2+(70/A))*20
1130 IF a<=16 THEN e=((ABS(y))+a*4)*1.2
1140 IF E>700 THEN E=790
1150 IF x>=320 THEN x=320
1160 IF y<=580 THEN y=-750
1170 IF a<=0.1 THEN y=-1000:e=-1050
1180 xi=3*a+360
1190 yi=360-INT(LOG(a+0.6)*72)
1200 MOVE xi,yi:FOR o=0 TO 360 STEP 4 :DRAW xi+40*COS(o),yi+40*SIN(o),2,0:NEXT
1210 MOVE xi,yi-30:FILL 2
1220 MOVE x,y:FOR o=0 TO 360 STEP 8 :DRAW x+e*COS(o),y+e*SIN(o),1,3:NEXT
1230 x1=x:y1=y
1240 IF y<=1 THEN y1=4
1250 MOVE x1,y1-3:FILL 1
1260 y2=416-(INT(145-(LOG(a+0.5)*34))*2)
1270 x2=(a*5)+250
1280 GRAPHICS PEN 1,1:MOVE x2,y2-4,,1:PRINT CHR$(250);
1290 GRAPHICS PEN 1,1:MOVE x2,y2-20,,1:PRINT CHR$(251);
1300 IF a<=0.1 AND FNA(v*3600)>85 THEN 1390
1310 IF c=0 THEN 1350
1320 GRAPHICS PEN 3,1:MOVE x2+2,y2-38,,1:PRINT CHR$(252);
1330 SOUND 1,1000,0,12,2
1340 SOUND 2,900,0,12,2
1350 FOR i=1 TO 2500:NEXT
1360 MODE 1:BORDER 13
1370 INK 0,13:INK 1,0
1380 TAGOFF:RETURN
1390 GRAPHICS PEN 8,1:MOVE x2+16,y2-20,3,0:FOR o=0 TO 360 STEP 7 :DRAW x2+16+45*
COS(o),y2-40+45*SIN(o),,0:NEXT
1400 MOVE x2-5,y2-10,,1:FILL 3
1410 FOR i=1 TO 300
1420 SOUND 135,100,0,13,3,1,20
1430 NEXT
1440 GRAPHICS PEN 6,1:MOVE x2+16,y2-20,0,2:FOR o=0 TO 360 STEP 7 :DRAW x2+16+45*
COS(o),y2-40+45*SIN(o):NEXT
1450 MOVE x2-5,y2-10:FILL 0
1460 GOTO 1350
```

DEMOLER



Esta es una versión para AMSTRAD del juego del muro. Tienes ante tí una pared compuesta por tres filas de ladrillos radiactivos que debes intentar derribar. Dispones para ello de 5 bolas. Según vayas practicando, podrás ir aumentando el nivel de dificultad del juego.


```
10 SYMBOL 250,&X100000,&X1110000,&X1110000,&X1110000,&X100000,&X0,&X0,&X0
20 INK 0,1:INK 1,26:INK 13,6:INK 3,2,8:PAPER 0
30 MODE 0:BORDER 1:LOCATE 1,12:PEN 3:PRINT "      DEMOLICION      ":FOR t=10 TO 500 ST
EP 10:SOUND 5,t,10:NEXT
40 MODE 1:PEN 1:PRINT:PRINT:PRINT "      ESTAS ANTE UNA PARED RADIOACTIVA  "
50 PRINT:PRINT "      Y TIENES 5 BOLAS PARA DERRIBARLA"
60 PRINT:PRINT:PRINT:PRINT "      !! BUENA SUERTE !!"
70 PRINT:PRINT:PRINT:PRINT "      (UTILIZA JOYSTICK O CURSOR)"
80 LOCATE 1,20:PRINT"PARA COMENZAR EL JUEGO PULSA UNA TECLA"
90 IF INKEY$="" THEN 90 ELSE CLS
100 t=0
110 MODE 0:BORDER 10
120 INK 0,26:INK 1,1:INK 2,2:INK 4,0:INK 8,26:INK 5,22
130 FOR s=20 TO 640 STEP 55
140 MOVE s-20,390,,1:GRAPHICS PEN 2,1:DRAW s+20,390,,1:DRAW s+20,370,,1:DRAW s-2
0,370,,1:DRAW s-20,390,,1:MOVE s,380,,1:FILL 3
150 IF s>570 THEN 170
160 MOVE s,365,,1:GRAPHICS PEN 2,1:DRAW s+40,365,,1:DRAW s+40,345,,1:DRAW s,345,
,1:DRAW s,365,,1:MOVE s+10,355,,1:FILL 3
170 MOVE s-20,340,,1:GRAPHICS PEN 2,1:DRAW s+20,340,,1:DRAW s+20,320,,1:DRAW s-2
0,320,,1:DRAW s-20,340,,1:MOVE s,330,,1:FILL 3:NEXT
180 a$=CHR$(136)+CHR$(132)
190 B$=CHR$(250)
200 TAG
210 j=300:z=20
220 xi=17:yi=12
230 x1=INT(RND*10)*64:y1=315
240 GRAPHICS PEN 4,1:MOVE j,z,,0:PRINT a$;
250 x=x1-xi:y=y1-yi
260 IF x>=640 OR x<=-10 THEN xi=-xi:GOTO 240
270 x1=x:y1=y
280 IF y>=400 THEN yi=-yi:GOTO 240
290 IF y<=0 THEN 500
300 GRAPHICS PEN 1,1:MOVE x,y,,3:PRINT B$;
310 IF INKEY(8)<>-1 OR INKEY(74)<>-1 THEN c1=1:c4=1
320 IF INKEY(1)<>-1 OR INKEY(75)<>-1 THEN c1=1:c5=1
330 IF TEST(X-2,Y)=3 AND Y<=340 AND YI<0 THEN H=X-2:R=1:GOTO 390
340 IF TEST(X+14,Y+2)=3 AND Y<=340 AND YI<0 THEN H=X+14:R=1:GOTO 390
350 IF TEST (X-2,Y)=3 AND Y>=345 AND Y<=365 AND YI<0 THEN H=X-2 :R=2:GOTO 390
360 IF TEST (X+14,Y+2)=3 AND Y>=345 AND Y<=365 AND YI<0 THEN H=X+14 :R=2:GOTO 39
0
370 IF TEST (X-2,Y)=3 AND Y>365 AND YI<0 THEN H=X-2 :R=3
380 IF TEST (X+14,Y+2)=3 AND Y>365 AND YI<0 THEN H=X+14 :R=3 ELSE GOTO 430
390 b=INT(h/55)*55
400 IF R=1 THEN MOVE b,340,,0:GRAPHICS PEN 8,1:DRAW b+40,340,,0:DRAW b+40,320,,2
:DRAW b,320,,2:DRAW b,340,,2:MOVE b+10,330,,2:FILL 0
:YI=-YI:TAGOFF:PRINT CHR$(7);:t=t+10:TAG
410 IF R=2 THEN MOVE b+20,365,,0:GRAPHICS PEN 8,1:DRAW b+60,365,,0:DRAW b+60,345
,,2:DRAW b+20,345,,2:DRAW b+20,365,,2:MOVE b+30,355,
,2:FILL 0:YI=-YI:TAGOFF:PRINT CHR$(7);:t=t+20:TAG
420 IF R=3 THEN MOVE b,390,,0:GRAPHICS PEN 8,1:DRAW b+40,390,,0:DRAW b+40,370,,2
:DRAW b,370,,2:DRAW b,390,,2:MOVE b+10,380,,2:FILL 0
:YI=-YI:TAGOFF:PRINT CHR$(7);:t=t+50:TAG
430 IF TEST (X,Y-8)=4 THEN yi=-yi:GOTO 450
440 IF TEST (X+16,Y-8)=4 THEN yi=-yi
450 IF c1=1 THEN GRAPHICS PEN 3,1:MOVE j,z,,2:PRINT a$:c1=0
460 IF c4=1 THEN j=j-12:c4=0
```



```

470 IF c5=1 THEN j=j+12:c5=0
480 GRAPHICS PEN 6,1:MOVE x,y,,2:PRINT B$;
490 GOTO 240
500 f=f+1:q=0
510 FOR I=1 TO 640 STEP 5
520 IF TEST(I,360)=3 OR TEST(I,355)=3 OR TEST(I,330)=3 THEN q=1:GOTO 540
530 NEXT
540 IF q=0 THEN l=1:GOTO 620
550 IF q=1 AND f>4 THEN e=1:GOTO 620
560 GRAPHICS PEN 3,1:MOVE j,z,,2:PRINT a$:c1=0
570 TAGOFF
580 FOR i=1 TO 40
590 PRINT CHR$(7);:NEXT
600 TAG
610 GOTO 210
620 MODE 1
630 IF l=1 THEN 650
640 IF e=1 THEN 660
650 LOCATE 1,10:PRINT" *** ENHORABUENA LO CONSEGUISTES ***":GOTO 670
660 LOCATE 1,10:PRINT" **** FIN DE LA PARTIDA ****"
670 PRINT:PRINT" ***** PUNTOS=";t;"*****"
680 PRINT:PRINT" ** QUIERES JUGAR OTRA VEZ iS/N? **"
690 S$=INKEY$
700 IF S$="S" OR S$="s" THEN f=0:CLR:GOTO 110
710 IF S$="N" OR S$="n" THEN CLS:END
720 GOTO 690

```



SUB-85

Tu submarino va equipado con un sistema de seguridad para ahorrar torpedos, que sólo te permite disparar cuando hayas acertado con las coordenadas exactas del buque enemigo.

Los instrumentos de abordo te indican con una flecha, al apretar el disparador, a la distancia a la que estás disparando, pero del ángulo sólo te indican si es 0 o si es positivo (derecha) o negativo (izquierda) con respecto a la vertical de tu submarino.


Debes localizar el valor exacto del ángulo lo antes posible y esquivar la artillería enemiga.

```

10 CLEAR
20 INK 0,1:INK 1,26:INK 3,10,12
30 MODE 0:BORDER 25:LOCATE 1,13:PEN 3:PRINT" SUBMARINO-85 ":FOR t=10 TO 500
STEP 10:SOUND 5,t,10:NEXT
40 MODE 1:BORDER 1:PEN 1:PRINT" Tu submarino va equipado con un"
50 PRINT" dispositivo de seguridad,para ahorrar"
60 PRINT" torpedos, que solo dispara cuando tu"
70 PRINT" hayas acertado con las coordenadas"
80 PRINT" exactas del buque enemigo."
90 PRINT:PRINT" Los instrumentos de abordo te in-"
100 PRINT" dican con una flecha, al apretar el"
110 PRINT" disparador, a que distancia estas ti-"
120 PRINT" rando; pero del angulo solo te dicen"

```

CINTA "TSA 4TB
DISCO 1CB "SUB85" SK




```
130 PRINT" si es 0 o es positivo(derecha) o ne-"
140 PRINT" gativo(izq.) con respecto a la ver-"
150 PRINT" tical de tu submarino."
160 PRINT:PRINT" Calcular su valor exacto es mision"
170 PRINT" del Capitan:.... !! TU MISION !!"
180 PRINT:PRINT" Utiliza el cursor para moverte y"
190 PRINT" variar la distancia, i< y i> para"
200 PRINT" variar el angulo y dispara con COPY"
210 PRINT:PRINT:PRINT" (TAMBIEN PUEDES USAR JOYSTICK)"
220 LOCATE 1,25:PRINT" PARA COMENZAR EL JUEGO PULSA UNA TECLA"
230 IF INKEY$="" THEN 230 ELSE CLS
240 MODE 1:DEG:INK 3,11:INK 1,26:BORDER 1
250 TAG
260 SYMBOL 250,0,8,8,255,123,15,1,0
270 SYMBOL 251,14,15,11,255,187,255,255,31
280 SYMBOL 252,0,14,144,255,6,204,248,240
290 SYMBOL 245,3,33,32,33,255,237,127,31
300 SYMBOL 246,64,241,193,226,255,173,255,255
310 SYMBOL 247,128,224,135,200,255,102,252,248
320 SYMBOL 254,64,64,64,64,64,160,160
330 SYMBOL 255,&X0,&X0,&X11000000,&X11111100,&X11000000
340 sub$= CHR$(250)+CHR$(251)+CHR$(252)
350 bar$= CHR$(245)+CHR$(246)+CHR$(247)
360 torp$=CHR$(254)
370 bom$=CHR$(255)
380 MOVE 0,375:DRAW 640,375,11
390 MOVE 0,302:DRAW 640,302,11
400 MOVE 1,365,,1:FILL 11
410 d=286:YT=40
420 MOVE 1,395,1:PRINT "ANGULO=";a;:MOVE 190,395: PRINT"DISTANCIA=";d;:MOVE 450,
395:PRINT"PUNTOS=";P;
430 sx=300:sy=30:by=316:y5=310:x5=-20
440 x6=-20:y6=310: x7=-20:y7=310
50 MOVE bx,by,1,1:PRINT bar$;
460 IF jr=0 THEN MOVE sx,sy:PRINT sub$;:jr=2
470 r=r+1
480 IF z<1 THEN 520
490 IF z1>=3 THEN z1=0
500 z1=z1+1:r=1:TAGOFF:PRINT CHR$(7);:TAG
510 ON z1 GOSUB 1200,1230,1260
520 z=r/11
530 IF h5=1 THEN x5=x5+x9:y5=y5-9
540 IF y5<=35 AND x5>6>sx AND x5<sx+48 AND y5>16 THEN 1080
550 MOVE x5,y5:PRINT bom$;
560 IF h6=1 THEN x6=x6+x10:y6=y6-9
570 IF y6<=35 AND x6>6>sx AND x6<sx+48 AND y6>16 THEN 1080
580 MOVE x6,y6:PRINT bom$;
590 IF h7=1 THEN x7=x7+x11:y7=y7-9
600 IF y7<=35 AND x7>6>sx AND x7<sx+48 AND y7>16 THEN 1080
610 MOVE x7,y7:PRINT bom$;
620 IF H=0 THEN 680
630 XT=XT+C1:YT=YT+6
640 IF XT>=BX+12 AND XT<=BX+36 THEN XT=BX+24
650 IF YT>=295 AND XT=BX+24 THEN P=P+10:CBX=1:XT=700:H=0:YT=40:TAGOFF:PRINT CHR$
(7);:TAG:SOUND 7,4000,5,15,0,0,5:GOSUB 910
660 IF yt>300 THEN yt= 40:h=0
670 MOVE XT,YT:PRINT TORP$;:HR=1
680 IF sx>bx THEN x2=sx-bx:c1=-6 ELSE x2=bx-sx:c1=6
690 IF H2=1 THEN MOVE x4+12,300,,1:PRINT CHR$(94);:H2=0
700 IF INKEY(0)<>-1 OR INKEY(72)<>-1 THEN d=d+2:h4=1:GOSUB 950
710 IF INKEY(2)<>-1 OR INKEY(73)<>-1 THEN d=d-5:h4=1:GOSUB 950
720 IF INKEY(8)<>-1 OR INKEY(74)<>-1 THEN csx=-1:jr=1
730 IF INKEY(1)<>-1 OR INKEY(75)<>-1 THEN csx=+1:jr=1
740 IF HR=1 THEN MOVE XT,YT,,1:PRINT TORP$;: HR=0
750 IF INKEY(76)<>-1 OR INKEY(9)<>-1 THEN h4=0:GOSUB 980
760 a$=INKEY$
770 IF a$="," THEN a=a-2:GOSUB 930
780 IF a$="." THEN a=a+2:GOSUB 930
790 IF jr=1 THEN MOVE sx,sy,,1:PRINT sub$;:jr=0
800 IF csx=1 THEN sx=sx+4:csx=0
810 IF csx=-1 THEN sx=sx-4:csx=0
820 IF sx>640 OR sx<0 THEN sx=0
830 IF h5=1 THEN MOVE x5,y5,,1:PRINT bom$;
840 IF h6=1 THEN MOVE x6,y6:PRINT bom$;
850 IF h7=1 THEN MOVE x7,y7:PRINT bom$;
860 MOVE bx,by,,1:PRINT bar$;
870 IF CBX=1 THEN BX=0:CBX=0
880 bx=bx+3
890 IF bx>640 THEN bx=0
```



```

900 GOTO 450
910 MOVE 560,395,,0:PRINT P;
920 RETURN
930 MOVE 116,395,,0:PRINT a;
940 RETURN
950 IF d<=286 THEN d=286
960 MOVE 350,395,,0:PRINT d;
970 RETURN
980 Dr=INT(SQR(x2^2+286^2))
990 ar=x2/dr
1000 x3=INT(SQR(d^2-286^2))
1010 IF a=0 THEN x4=sx:GOTO 1050
1020 IF a<0 THEN x4=sx-x3:GOTO 1050
1030 IF a>0 THEN x4=sx+x3
1040 IF h4=0 AND d-4<=dr AND d+6>=dr AND SIN(a-2)<=ar AND SIN(a+2)>=ar AND sx<=b
x THEN h=1:xt=sx:TAGOFF:PRINT CHR$(7):TAG:GOTO 1060
1050 IF h4=0 AND d-4<=dr AND d+6>=dr AND SIN(a-2)<=-ar AND SIN(a+4)>=-ar AND sx>
=bx THEN h=1:xt=sx:TAGOFF:PRINT CHR$(7):TAG
1060 MOVE x4+12,300,,1:PRINT CHR$(94);H2=1:H4=0:FRAME
1070 RETURN
1080 FOR i=1 TO 30
1090 SOUND 7,4000,5,15,0,0,5:NEXT
1100 MODE 1
1110 LOCATE 1,10:PRINT "      LO SIENTO HAS SIDO ALCANZADO      "
1120 PRINT:PRINT "      TUS PUNTOS HAN SIDO :";P;      "
1130 PRINT:PRINT "      LO INTENTAS DE NUEVO (S/N) "
1140 B$= INKEY$
1150 IF B$="S" OR B$="s" THEN CLEAR:TAG :GOTO 240
1160 IF B$="N" OR B$="n" THEN GOTO 1180
1170 GOTO 1140
1180 CLEAR:CLS
1190 END
1200 x5=bx+14:h5=1:x9=x2/30:y5=300
1210 IF bx>sx THEN x9=-x9
1220 RETURN
1230 x6=bx+14:h6=1:x10=x2/30:y6=300
1240 IF bx>sx THEN x10=-x10
1250 RETURN
1260 x7=bx+14:h7=1:x11=x2/30:y7=300
1270 IF bx>sx THEN x11=-x11
1280 RETURN

```



PIN-PONG


En este juego de tenis de mesa tu oponente es un difícil rival al que no te será fácil vencer, sobre todo en los niveles altos de dificultad. Si quieres tener alguna oportunidad ante tu oponente debes hacer que éste suba a la red y cambiarle entonces la bola de un lado a otro de la mesa a fin de cojerle a contrapie.

```

10 SYMBOL 250,&X100000,&X1110000,&X1110000,&X1110000,&X100000,&X0,&X0,&X0
20 INK 0,3:INK 1,26:INK 13,6:INK 3,9,13:PAPER 0
30 MODE 0:BORDER 13:LOCATE 1,12:PEN 3:PRINT"      PING-PONG      ":FOR t=10 TO 500 S
TEP 10:SOUND 5,t,10:NEXT
40 MODE 1:PEN 1:PRINT:PRINT:PRINT"      BORIS AMSKER ES UN RIVAL MUY"
50 PRINT:PRINT "      DIFICIL DE SORPRENDER PERO CON"
60 PRINT:PRINT"      RAPIDEZ Y ASTUCIA PODRAS VENCERLE"
70 PRINT:PRINT:PRINT:PRINT "      !! BUENA SUERTE !!"

```

CINTA "TSA UTB
DISCO 1CB "PINPONG" 4K



```
80 PRINT:PRINT:PRINT:PRINT"          (UTILIZA JOYSTICK O CURSOR)"
90 LOCATE 1,20:PRINT" PARA COMENZAR EL JUEGO PULSA UNA TECLA"
100 IF INKEY$="" THEN 100 ELSE CLS
110 CLS:PRINT CHR$(7)::LOCATE 1,10:LINE INPUT " NOMBRE DEL JUGADOR:";Q$
120 CLS:PRINT CHR$(7)::LOCATE 6,7:PRINT "*** ELIJE NIVEL DE JUEGO ***":LOCATE 12
,9:LINE INPUT " DE 10 A 18:";R$
130 R=VAL(R$)
140 IF R<10 OR R>18 THEN 120
150 CLS:PRINT CHR$(7)::LOCATE 1,10:LINE INPUT " CUANTAS BOLAS SE VAN A JUGAR";T$


160 T=VAL(T$)
170 IF T<=0 OR T>30 THEN 150
180 CLS:LOCATE 15,8:PRINT Q$
190 LOCATE 6,10:PRINT" JUEGAS EN EL LADO IZQUIERDO"
200 LOCATE 9,12:PRINT" PARA MOVER TU RAQUETA"
210 LOCATE 7,14:PRINT" UTILIZA JOYSTICK O CURSOR"
220 LOCATE 5,18:PRINT" PULSA UNA TECLA PARA CONTINUAR"
230 IF INKEY$="" THEN 230 ELSE CLS
240 FOR I=1 TO 25:PRINT CHR$(7):NEXT
250 A=10
260 MODE 0:BORDER 6:INK 0,23:INK 1,0:INK 8,23:INK 5,6
270 X=320:Y=RND*380:P=1
280 X1=A:Y1=R
290 J=12:K=200
300 L=610:M=200
310 TAG
320 IF X>305 AND X<335 THEN GRAPHICS PEN 3,1:MOVE 320,400:PLOT 320,400:DRAW 320,
0,,0
330 X1=X-X1
340 Y1=Y+Y1
350 X=X1:Y=Y1
360 GRAPHICS PEN 1,1:MOVE X,Y:PRINT CHR$(250);
370 IF C1=0 THEN GRAPHICS PEN 1,1:MOVE J,K:PRINT CHR$(138)::MOVE J,K+16:PRINT CH
R$(138);
380 IF P=1 THEN GRAPHICS PEN 1,1:MOVE L,M:PRINT CHR$(133)::MOVE L,M+16,,0:PRINT
CHR$(133);:P=0
390 FRAME
400 IF Y>=400 OR Y<=0 THEN Y1=-Y1
410 IF INKEY(0)<>-1 OR INKEY(72)<>-1 THEN C1=1:D1=1
420 IF INKEY(2)<>-1 OR INKEY(73)<>-1 THEN C1=1:D2=1
430 IF INKEY(8)<>-1 OR INKEY(74)<>-1 THEN C1=1:C4=1
440 IF INKEY(1)<>-1 OR INKEY(75)<>-1 THEN C1=1:C5=1
450 IF X1<0 THEN 500
460 IF TEST (X+1,Y+16)=1 THEN X1=-X1:GOTO 530
470 IF TEST (X,Y-8)=1 THEN X1=-X1:GOTO 530
480 IF TEST (X+1,Y-1)=1 THEN X1=-X1:GOTO 530
490 IF X1>0 THEN 530
500 IF TEST (X+18,Y)=1 THEN X1=-X1:P=1:GOTO 530
510 IF TEST (X+18,Y+16)=1 THEN X1=-X1:P=1:GOTO 530
520 IF TEST (X+18,Y+8)=1 THEN X1=-X1:P=1
530 IF C1=1 THEN GRAPHICS PEN 8,1:MOVE J,K,0:PRINT CHR$(138)::MOVE J,K+16,,0:PR
INT CHR$(138);:C1=0
540 IF D1=1 THEN K=K+12:D1=0
550 IF D2=1 THEN K=K-12:D2=0
560 IF C4=1 THEN J=J-12:C4=0
570 IF C5=1 THEN J=J+12:C5=0
580 IF J<=-20 THEN J=-20
590 IF J>=290 THEN J=290
600 IF K>=400 THEN K=400
610 IF K<=0 THEN K=0
620 IF X>645 THEN A=-10:AT=AT+1:T=T-1:GOTO 670
630 IF X<-10 THEN A=10:BT=BT+1:T=T-1:GOTO 670
640 IF X1<0 THEN GOSUB 860
650 GRAPHICS PEN 8,1:MOVE X,Y,0:PRINT CHR$(250);
660 GOTO 320
670 TAGOFF
680 FOR I=1 TO 40
690 PRINT CHR$(7):NEXT
700 TAG
710 X=320
720 MODE 1
730 IF T>0 THEN 770
740 IF BT>=AT THEN 760
750 LOCATE 1,10:PRINT" *** ENHORABUENA LO CONSEGUISTES ***"
760 LOCATE 3,7:PRINT" ***** FIN DEL PARTIDO *****"
770 LOCATE 8,13:PRINT"***** ";Q$;"=";JAT;"*****"
780 LOCATE 7,16:PRINT"***** Boris AMSker=";BT;"*****"
790 FOR I=1 TO 2000:NEXT
800 IF T>0 THEN 260
```



```

810 PRINT:PRINT:PRINT"  ** QUIERES JUGAR OTRA VEZ iS/N¿ **"
820 S$=INKEY$
830 IF S$="S" OR S$="s" THEN f=0:CLER:GOTO 110
840 IF S$="N" OR S$="n" THEN CLS:END
850 GOTO 820
860 IF Y1>0 AND M+6<Y THEN vm=1
870 IF Y1<0 AND M+12>Y THEN vm=-1
880 IF X<285 AND L>340 AND J<255 THEN u1=-1
890 IF X>L-30 AND L<620 AND Y<M AND Y>M-200 THEN u1=1:vm=-1
900 IF X>L-30 AND L<620 AND Y>M+32 AND Y<M+200 THEN uL=1:vm=1
910 IF x>270 AND l<580 AND j>255 THEN u1=1
920 IF vm<>0 OR u1<>0 THEN GRAPHICS PEN 8,1:MOVE 1,m,,0:PRINT CHR$(133);:MOVE 1,
m+16,,0:PRINT CHR$(133);:p=1
930 IF vm=-1 THEN m=m-r
940 IF vm=1 THEN m=m+r
950 IF u1=-1 THEN l=l-r
960 IF u1=1 THEN l=l+9
970 vm=0:u1=0
980 P=1
990 RETURN

```



TANK-2000

Con tu tanque te enfrentas a un enemigo que aunque más numeroso posee un armamento muy inferior al tuyo. De todos modos y dada su superioridad numérica deberás estar muy atento y aguardar la serenidad a fin de no ser destruido.


Como referencia tienes una línea en la pantalla la cual no debe ser rebasada por ningún tanque enemigo o te verás en apuros.

```


10 INK 0,1:INK 1,26:INK 2,6:INK 3,2,8:PAPER 0
20 MODE 0:BORDER 1:LOCATE 1,12:PEN 3:PRINT"  TANQUE-2000  ":FOR t=10 TO 500 S
TEP 10:SOUND 5,t,10:NEXT
30 MODE 1:PEN 1:PRINT:PRINT"  Los viejos tanques enemigos avanzan"
40 PRINT:PRINT"  hacia tu territorio, aprovecha"
50 PRINT:PRINT"  el mayor alcance de tus armas y..."
60 PRINT:PRINT:PRINT"  NO DEJES QUE SE PASEN DE LA RAYA"
70 PRINT:PRINT:PRINT"  Utiliza el cursor para moverte "
80 PRINT"  y dispara con COPY"
90 PRINT:PRINT:PRINT"  (TAMBIEN PUEDES USAR JOYSTICK)"
100 LOCATE 1,20:PRINT"  PARA COMENZAR EL JUEGO PULSA UNA TECLA"
110 IF INKEY$="" THEN 110 ELSE CLS
120 SYMBOL 241,&X10110,&X10110,&X10110,&X0,&X111,&X1010,&X1101,&X11111111
130 SYMBOL 242,&X11011011,&X11011011,&X11011011,&X1110000,&X11011100,&X101110,&X
1110011,&X10011010
140 SYMBOL 243,&X11111111,&X1101,&X1010,&X111,&X0,&X10110,&X10110,&X10110
150 SYMBOL 244,&X10011010,&X11110011,&X101110,&X1101100,&X1110000,&X11011011,&X
11011011,&X11011011
160 SYMBOL 245,&X0,&X0,&X0,&X0,&X0,&X11,&X11110,&X11
170 SYMBOL 250,&X0,&X0,&X11000000,&X1111000,&X11000000,&X0,&X0,&X0
180 SYMBOL 252,&X11011011,&X11011011,&X11011011,&X11110,&X110011,&X1101101,&X110
11110,&X1010011
190 SYMBOL 253,&X1010011,&X11011110,&X1101101,&X110011,&X11110,&X11011011,&X1101
1011,&X11011011
200 SYMBOL 254,&X1101000,&X1101000,&X1101000,&X0,&X11100000,&X10110000,&X1010000
,&X11111111
210 SYMBOL 255,&X11111111,&X1010000,&X10110000,&X11100000,&X0,&X1101000,&X110100
0,&X1101000

```

CINTA "TSA 4TB
DISCO 1CB "TANK2000" 6k



```
220 INK 0,12
230 INK 1, 26
240 INK 2,22
250 INK 3,6
260 INK 4,6
270 INK 6,0
280 INK 7,9
290 INK 8,20
300 t5=-45:t6=330:t7=330
310 CLG:MODE 0 :DEG:ORIGIN 0,40:MOVE 0,40
320 BORDER 6
330 PLOT 0,320
340 DRAW 0,360,8,0:DRAW 0,320,8,3:DRAW 640,320,8,3
350 MOVE 525,330,8,2:FILL 8
360 PLOT 320,290
370 DRAW 320,290,4,1:DRAW 320,0,4,1
380 PRINT:PRINT " PUNTOS="
390 pr$=CHR$(252)+CHR$(254)
400 pl$=CHR$(253)+CHR$(255)
410 pb$ =CHR$(250)
420 pm$=CHR$(43)
430 te$=CHR$(241)+CHR$(242)
440 ta$=CHR$(243)+CHR$(244)
450 be$=CHR$(245)
460 GOSUB 1040
470 TAG
480 xc=20:yc=175
490 x=xc+100:y=yc-8
500 K=1:r=1
510 IF K=1 THEN GRAPHICS PEN 4,1:MOVE xc,yc-16,,0:PRINT pl$;:MOVE xc,yc,,0:PRIN
T pr$;:K=0
520 h=h-5.75:h1=h1-6:h2=h2-5.5
530 GRAPHICS PEN 8,1:MOVE h,t,,3:PRINT ta$;:MOVE h,t+16,,3:PRINT te$;
540 GRAPHICS PEN 8,1:MOVE h1,t1,,3:PRINT ta$;:MOVE h1,t1+16,,3:PRINT te$;
550 GRAPHICS PEN 8,1:MOVE h2,t2,,3:PRINT ta$;:MOVE h2,t2+16,,3:PRINT te$;
560 IF r=1 THEN GRAPHICS PEN 6,1:MOVE x,yc-8,,0:PRINT pm$;:r=0
570 IF kb=1 THEN GRAPHICS PEN 6,1:MOVE h5,t5,,3:PRINT be$;
580 IF kc=1 THEN GRAPHICS PEN 6,1:MOVE h6,t6,,3:PRINT be$;
590 IF kd=1 THEN GRAPHICS PEN 6,1:MOVE h7,t7,,3:PRINT be$;
600 IF kb=0 AND h<320 AND h>85 THEN kb=1:h5=h-32:t5=t+14
610 IF kc=0 AND h1<320 AND h1>85 THEN kc=1:h6=h1-32:t6=t1+14
620 IF kd=0 AND h2<320 AND h2>85 THEN kd=1:h7=h2-32:t7=t2+14
630 IF h<50 OR h1<50 OR h2<50 THEN 1140
640 IF INKEY(72)<>-1 OR INKEY(0)<>-1 THEN r=1:ho=1:k=1
650 IF INKEY(73)<>-1 OR INKEY(2)<>-1 THEN r=1:hu=1:k=1
660 IF INKEY(74)<>-1 OR INKEY(8)<>-1 THEN mu=1:r=1
670 IF INKEY(75)<>-1 OR INKEY(1)<>-1 THEN mo=1:r=1
680 IF ba=0 AND INKEY(76)<>-1 OR ba=0 AND INKEY(9)<>-1 THEN ba=1:xa=xc+56
690 IF ba=1 THEN xa=xa+12:GRAPHICS PEN 8,1 :MOVE xa,yc-8:PRINT pb$;
700 IF ba=1 AND xa>x AND xa>h AND xa<h+64 AND yc>t AND yc<t+32 THEN uf=0:GOSUB 9
90
710 IF ba=1 AND xa>x AND xa>h1 AND xa<h1+64 AND yc>t1 AND yc<t1+32 THEN v=0:GOS
UB 990
720 IF ba=1 AND xa>x AND xa>h2 AND xa<h2+64 AND yc>t2 AND yc<t2+32 THEN w=0:GOSU
B 990
730 IF xa>x THEN GOSUB 1040
740 IF kb=1 THEN GRAPHICS PEN 8,1 :MOVE h5,t5,,2:PRINT be$;
750 IF kc=1 THEN GRAPHICS PEN 8,1 :MOVE h6,t6,,2:PRINT be$;
760 IF kd=1 THEN GRAPHICS PEN 8,1 :MOVE h7,t7,,2:PRINT be$;
770 IF kb=1 THEN h5=h5-10
780 IF kc=1 THEN h6=h6-10
790 IF kd=1 THEN h7=h7-10
800 IF h5<-10 THEN kb=0
810 IF h6<-10 THEN kc=0
820 IF h7<-10 THEN kd=0
830 IF H5<50 AND H5>5 AND t5<yc+16 AND t5>yc-24 THEN 1140
840 IF H6<50 AND H6>5 AND t6<yc+16 AND t6>yc-24 THEN 1140
850 IF H7<50 AND H7>5 AND t7<yc+16 AND t7>yc-24 THEN 1140
860 IF ba=1 THEN GRAPHICS PEN 7,1:MOVE xa,yc-8,,2:PRINT pb$;
870 IF k=1 THEN GRAPHICS PEN 3,0:MOVE xc,yc,,2:PRINT pr$;:MOVE xc,yc-16,,2:PRINT
pl$;:k=1
880 IF R=1 THEN GRAPHICS PEN 8,1 :MOVE x,yc-8,,2:PRINT pm$;
890 IF mo=1 THEN x=x+12:mo=0
900 IF mu=1 THEN x=x-12:mu=0
910 IF ho=1 THEN yc=yc+10:ho=0
920 IF hu=1 THEN yc=yc-10:hu=0
930 IF yc<=0 THEN yc=0
940 IF yc>=310 THEN yc=310
950 GRAPHICS PEN 7,1:MOVE h,t,,2:PRINT ta$;:MOVE h,t+16,,2:PRINT te$;
```

```
960 GRAPHICS PEN 7,1:MOVE h2,t2,,2:PRINT ta$;:MOVE h2,t2+16,,2:PRINT te$;
970 GRAPHICS PEN 7,1:MOVE h1,t1,,2:PRINT ta$;:MOVE h1,t1+16,,2:PRINT te$;
980 GOTO 510
990 SOUND 7,4000,5,15,0,0,5
1000 GRAPHICS PEN 7,1:MOVE h,t,,2:PRINT ta$;:MOVE h,t+16,,2:PRINT te$;
1010 GRAPHICS PEN 7,1:MOVE h1,t1,,2:PRINT ta$;:MOVE h1,t1+16,,2:PRINT te$;
1020 GRAPHICS PEN 7,1:MOVE h2,t2,,2:PRINT ta$;:MOVE h2,t2+16,,2:PRINT te$;
1030 pt=pt+10
1040 TAGOFF
1050 PRINT CHR$(7);
1060 LOCATE 9,2:PRINT ;pt;
1070 TAG
1080 IF uf=0 THEN t=RND*78:h=640:uf=1
1090 IF v=0 THEN t1=RND*78+110:h1=650:v=1
1100 IF w=0 THEN t2=RND*78+220:h2=650:w=1
1110 GRAPHICS PEN 7,1 :MOVE xa,yc-8,,2:PRINT pb$;
1120 xa=xc:ba=0
1130 RETURN
1140 FOR i=1 TO 20
1150 SOUND 7,4000,5,15,0,0,5:NEXT
1160 ENV 1,100,2,2
1170 ENT 1,100,2,2
1180 SOUND 1,284,200,1,1,1
1190 GRAPHICS PEN 4,1:MOVE xc,yc,,0:PRINT pr$;:MOVE xc,yc-16,,0:PRINT pl$;
1200 GRAPHICS PEN 3,0:MOVE xc,yc,,2:PRINT pr$;:MOVE xc,yc-16,,2:PRINT pl$;
1210 yc=yc-8
1220 IF yc<=0 THEN 1240
1230 GOTO 1190
1240 MODE 1:CLS
1250 LOCATE 1,10:PRINT "          LO SIENTO HAS SIDO ALCANZADO          "
1260 PRINT:PRINT "          TUS PUNTOS HAN SIDO :";PT;          "
1270 PRINT:PRINT:PRINT "          LO INTENTAS DE NUEVO (S/N)          "
1280 B$= INKEY$
1290 IF B$="S" OR B$="s" THEN CLEAR:TAG :GOTO 220
1300 IF B$="N" OR B$="n" THEN GOTO 1320
1310 GOTO 1280
1320 CLEAR:CLS
1330 END
```

Boletín de suscripción

A remitir a TODO SOBRE EL AMSTRAD. C/. Bailén, 20. 1.º Izda. 28005 Madrid

Deseo suscribirme a los 11 n.ºs de la revista TODO SOBRE EL AMSTRAD por sólo 3.500 pts

El importe lo haré efectivo:

- ☐ Por giro postal n.º
- ☐ Por talón nominativo adjunto.
- ☐ Contra reembolso a la recepción del primer ejemplar, más gastos de envío.


Deseo suscribirme a partir del n.º (inclusive).

Nombre y apellidos:

Domicilio:

Ciudad: Teléfono:

Fecha: Firma:



ARCHIVO

En este archivo numérico podrás almacenar tantos datos como te permita la memoria de tu ordenador.


Estos datos no pueden tener más de seis cifras cada uno.

El archivo viene presentado en páginas de 30 datos, los cuales pueden ser modificados tantas veces como se precise, antes e incluso después de haber realizado cualquier tipo de cálculo con ellos.

```
10 INK 0,1:INK 1,26:INK 2,6:INK 3,2,8:PAPER 0
20 MODE 0:BORDER 1:LOCATE 1,12:PEN 3:PRINT" ARCHIVO NUMERICO ":FOR t=10 TO 500
STEP 10:SOUND 5,t,10:NEXT
30 MODE 1:PEN 1:PRINT:PRINT:PRINT" En este archivo numerico podras "
40 PRINT " almacenar tantas cantidades como "
50 PRINT " te permita la memoria de tu orde-"
60 PRINT " nador. Estas cantidades no pueden"
70 PRINT " tener mas de seis cifras cada una, "
80 PRINT " es decir, el numero mayor almace-"
90 PRINT " nable es 999999. El archivo viene"
100 PRINT " presentado en paginas de 30 datos"
110 PRINT " los cuales pueden ser modificados"
120 PRINT " tantas veces como se precise antes"
```


CINTA "TSA 4A
DISCO "ARCHIVO" 1CB 4K

Todo sobre el
AMSTRAD



Bailén, 20 - 1º - Izda.



28007 MADRID



```


130 PRINT          " e incluso despues de relizar cual-"
140 PRINT          " quier tipo de calculo con ellos."
150 PRINT :PRINT          "      Como ejemplo de calculo hemos"
160 PRINT          "      puesto una estadistica de errores,"
170 PRINT          "      si bien tu puedes introducir otros."
180 LOCATE 1,22:PRINT"      PULSA UNA TECLA PARA COMENZAR"
190 IF INKEY$="" THEN 190 ELSE CLS
200 INPUT "NUM. DE DATOS A INTRODUCIR";D
210 DIM A(D)
220 FOR I=1 TO D
230 PRINT:PRINT "DATO NUMERO";I;"=";:INPUT A(I)
240 IF A(I)>999999 THEN PRINT CHR$(7): PRINT"MAXIMO VALOR ADMITIDO= 999999":GOTO
230
250 IF I=D THEN PRINT:PRINT:PRINT"TODO LOS DATOS HAN SIDO INTRODUCIDOS":PRINT C
HR$(7);:FOR H=1 TO 800:NEXT H
260 NEXT
270 CLS
280 NP=1
290 IF NP=0 THEN PRINT"***** NO HAY MENOS PAGINAS *****":NP=NP+1
300 IF NP*30-D>=30 THEN PRINT"***** NO HAY MAS PAGINAS *****":NP=NP-1
310 Z=(NP-1)*30
320 FOR I=Z+1 TO D
330 LOCATE 1,2:PRINT"NUM. DE DATOS ";D
340 X=INT(((I-6)/10) MOD 30)*12-(NP-1)*36
350 IF X<0 THEN X=0
360 Y=((I-1) MOD 10)+4
370 LOCATE 23,2:PRINT"PAG. NUM.";NP
380 LOCATE X+1,Y:PRINT I;"-";A(I);:B$=STR$(A(I)):PRINT STRING$(7-LEN(B$),32)
390 IF I=Z+30 THEN GOTO 420
400 IF I=D THEN LOCATE 5,16:PRINT"*** FIN DE LOS DATOS ***"
410 NEXT
420 LOCATE 1,16:PRINT"PULSE + PARA PAG. SIGUIENTE      "
430 LOCATE 1,18:PRINT"PULSE - PARA PAG. ANTERIOR"
440 LOCATE 1,20:PRINT"PULSE M PARA MODIFICAR DATOS";:PRINT STRING$(12,32);
450 LOCATE 1,22:PRINT"PULSE C SI LOS DATOS SON CORRECTOS      "
460 A$=INKEY$
470 IF A$="+" THEN CLS:PRINT CHR$(7);:NP=NP+1:GOTO 290
480 IF A$="-" THEN CLS:PRINT CHR$(7);:NP=NP-1:GOTO 290
490 IF A$="M" OR A$="m" THEN PRINT CHR$(7);:GOTO 520
500 IF A$="C" OR A$="c" THEN CLS:PRINT CHR$(7);:GOTO 570
510 GOTO 460
520 LOCATE 1,22:PRINT STRING$(36,32)
530 LOCATE 1,20:INPUT "INDIQUE NUM. DE DATO A MODIFICAR";I:IF I>NP*30 OR I<NP*30
-29 OR I>D THEN PRINT CHR$(7);:LOCATE 1,22:PRINT"EL
DATO NUM.";I;"NO ESTA EN PANTALLA":FOR V=1 TO 200:PRINT CHR$(7);:NEXT:GOTO 420
540 PRINT CHR$(7);:LOCATE 1,20:PRINT"EL DATO NR";I;"CONTIENE EL VALOR ";A(I);:PR
INT STRING$(4,32)
550 LOCATE 1,22:INPUT "INTRODUCCA EL NUEVO VALOR ";A(I):IF A(I)>999999 THEN PRIN
T CHR$(7);:LOCATE 1,22:PRINT STRING$(40,32);:GOTO 55
0
560 GOTO 290
570 CLS:PRINT CHR$(7);:PRINT "QUE CALCULO REALIZARA CON SUS DATOS":PRINT:PRINT"
-PULSE R PARA REVISAR DATOS "
580 PRINT:PRINT"-PULSE G PARA ESTADISTICA DE ERRORES"
590 B$=INKEY$
600 IF B$="G" OR B$="g" THEN CLS:GOTO 630
610 IF B$="R" OR B$="r" THEN CLS:GOTO 290
620 GOTO 590
630 PRINT CHR$(7):PRINT:PRINT"*** CALCULO ESTADISTICO DE ERRORES ***"
640 FOR I=1 TO d
650 S=S+A(I)
660 NEXT
670 M=S/D
680 FOR I=1 TO d
690 S2=S2+(A(I)-M)^2
700 NEXT
710 E=SQR(S2/(D*(D-1)))
720 LOCATE 3,6:PRINT"MEDIA=";M
730 LOCATE 3,9:PRINT"ERROR ABSOLUTO=";E
740 ER=E/M
750 LOCATE 3,12:PRINT"ERROR RELATIVO=";ER
760 TE=ER/100
770 LOCATE 3,15:PRINT" % DE ERROR=";TE
780 LOCATE 1,20:PRINT"DESEA REALIZAR ALGUN CALCULO MAS (S/N)"
790 A$=INKEY$
800 IF A$="S" OR A$="s" THEN S=0:S2=0:CLS:GOTO 570
810 IF A$="N" OR A$="n" THEN 830
820 GOTO 790
830 CLS:END

```





TECNICAS DE PROGRAMACION AVANZADA CON AMSTRAD

Tecnicas de Programación Avanzada con **AMSTRAD**



KEITH HOOK



El Amstrad CPC464 ha sido diseñado para permitir al programador tener acceso a un excelente y potente lenguaje de programación: **Locomotive Basic**.

Este libro lo he escrito con la esperanza de que sea útil a los programadores que quieren conseguir el máximo provecho de su ordenador. En esencia, este libro viene a ser un recetario de ideas desarrolladas en torno al **interpretador Basic** del Amstrad.

Este libro no está diseñado para enseñar al novato todos los aspectos de la programación en Basic, sino más bien para suplementar un nivel supuesto de conocimiento espigado del manual de instrucciones.

Si tienes como mínimo una somera familiaridad con Basic, este libro te permitirá progresar a través de numerosos ejemplos hasta un nivel avanzado de pro-

gramación, mostrándote cómo incorporar sencillas rutinas en código máquina a tus programas en Basic. No te preocupes si no comprendes todas las instrucciones que se describen — **penetra hasta el extremo más profundo**. La mejor manera de aprender es practicando los ejemplos; la comprensión vendrá posteriormente.

En las páginas de este libro encontrarás una cantidad considerable de material de consulta y referencia ante el que puedes decidir si estudiarlo ahora o en una fecha ulterior; y no necesitas graduarte en informática para aplicar los nuevos conocimientos que vayas adquiriendo.

El material de este libro te mostrará cómo atrapar las potentes rutinas que descansan en lo profundo de la «**mollera**» del Amstrad. La mayoría de dichas rutinas pueden ser aprovechadas desde el Basic, y te permiti-


rá producir programas que son una mixtura de código máquina y Basic. Esta manera híbrida de programar el Amstrad te permitirá realizar tareas desde Basic que pudieran parecerse imposibles — **Cargar un programa en código máquina en una Tabla y ejecutarlo; o poner bajo control del CPC464 hasta 8 figuras animadas**. Este libro te mostrará cómo lograrlo.

No hay nada difícil ni misterioso sobre los métodos sugeridos. Desde luego, hay reglas, y algunos de vosotros todavía no habéis intentado dar el paso de gigante para pasar más allá del Basic, pero con un poco de aplicación todas las cosas son posibles. Lo reitero, **lánzate a fondo a las entrañas del CPC464**.

Aunque no es la intención de este libro enseñarte **lenguaje de Ensamblaje**, sí te proporcionará una buena perspectiva, y espero que un buen empujón que te anime a investigar un mundo completamente nuevo en la programación.

Con el propósito de aprovechar las ventajas del **sistema operativo** del ordenador, debes poseer un buen conocimiento de cómo opera la máquina, y por lo tanto puede que algunos de vosotros piensen que los capítulos primeros son un poco elementales. Sin embargo, te insto a que los leas; nunca se sabe, pero puede que incluso descubras que no lo conocías en absoluto.

P.V.P. 1.600 ptas.



ta-ma

- La mayor variedad de libros de microinformática, capaces de satisfacer todas sus necesidades, ya sean profesionales, familiares, culturales...

- Todo tipo de información bibliográfica sobre microordenadores, desde AMSTRAD a Sinclair QL

TARJETA DE PEDIDO

--	--	--	--	--	--	--	--

Domicilio de envíos:

Nombre y apellidos
Domicilio N.º Piso Pta.
C. P. Ciudad Provincia

Ruego sírvanse remitirme **CONTRA REEMBOLSO** los siguientes libros:

Número	Cantidad	TÍTULO Y AUTOR	Importe


Fecha / /

Fdo.:

EL TECLADO DEL AMSTAD

495
Ptas

YA ESTA EN LOS QUIOSCOS




R. CARRALON

DISPONIBLE PARA ZX SPECTRUM

SOFTWARE

Sound-on-Sound
producida y distribuida por Iberofón, S. A.
Tel. 671.22.00 / 04 / 08 / 12 / 16

JUEGA CON EL FUTURO



¡¡¡NO LO SUEÑES!!! ¡JUEGALO!
SIENTE LA EMOCION DE LO DESCONOCIDO
CORRE TU PROPIO RIESGO

SALVA A TU COMPAÑERO/A ATRAPADO/A
REUNE LOS FRAGMENTOS DEL CUADRO
SON TU AMULETO

"BLACK JACK" ¡¡¡EL JUEGO DE CARTAS
MAS EMOCIONANTE!!!

¡¡¡CASINO ROYAL!!!
"LA EMOCION DEL JUEGO POR TODO LO ALTO"

¡¡¡POR FIN EN CASTELLANO!!!
LA PRIMERA COMEDIA MUSICAL EN VIDEO-JUEGO